Determinacy of Degenerate Equilibria with Linear Part $x'=y$, $y'=0$ Using MACSYMA

R. H. Rand and W. L. Keith

Department of Theoretical and Applied Mechanics
Cornell University
Ithaca, New York 14853

ABSTRACT

A MACSYMA program is presented which implements Takens's method for proving determinacy of a flow in the neighborhood of an equilibrium point by successive blowup transformations. The program is applied to nonlinear systems with linear part $x' = y$, $y' = 0$, and an examination of the pattern of the results reveals an infinite sequence of determinacy theorems. The results are applied to the behavior of van der Pol's equation at infinity.

INTRODUCTION

In the study of dynamical systems, one is often confronted with the problem of finding the behavior of a system in the neighborhood of an equilibrium position. In the case of two dimensions, the general system may be written

$$x' = u(x,y),\qquad y' = v(x,y), \tag{1}$$

where we assume $u(0,0) = v(0,0) = 0$, i.e. the origin is an equilibrium position. The usual approach to finding the local behavior near $(0,0)$ is to expand (1) in a Taylor series about the origin:

$$x' = ax + by + O(2),\qquad y' = cx + dy + O(2). \tag{2}$$

It is well known that if the linear part of (2) is hyperbolic, i.e. if the real parts

APPLIED MATHEMATICS AND COMPUTATION 21:1–19 (1987)

52 Vanderbilt Ave., New York, NY 10017 0096-3003/87/$03.50

of the eigenvalues are nonzero, then the $O(2)$ terms in (2) cannot influence the topological nature of the local flow.

In this paper we shall be interested in the case in which the equations (2) are of the form

$$x' = y + O(2), \qquad y' = O(2). \tag{3}$$

This is the case of a double zero eigenvalue in which only one eigenvector exists. In this case it is not sufficient to consider only the linear terms of (3) when finding the local behavior.

Takens [6] has investigated the system (3) using normal forms and blowup transformations. He showed that a near-identity transformation may always be found which reduces any system (3) to the normal form:

$$x' = y + b_2x^2 + b_3x^3 + b_4x^4 + b_5x^5 + \cdots, \qquad y' = a_2x^2 + a_3x^3 + a_4x^4 + a_5x^5 + \cdots. \tag{4}$$

Takens also showed that if the coefficient $a_2 \neq 0$, then in some neighborhood of the origin, the flow given by the system (4) is topologically equivalent to the flow given by the simplified system

$$x' = y, \qquad y' = a_2x^2. \tag{5}$$

That is, the topological nature of the local flow about the origin in (4) is *determined* by the system (5), in the case that $a_2 \neq 0$. (Note that although the coefficient b_2 is not important here, it does play an important role in the unfolding of the singularity (4); see [1]. We do not consider unfoldings of (4) in this paper.)

Our purpose in this paper is to generalize Takens's result for systems (4) in which the coefficient $a_2 = 0$. We use Takens's method of blowup transformations implemented on the computer algebra system MACSYMA. In the body of the paper, we explain Takens's method in detail, as well as providing the MACSYMA program in the Appendix. Now, however, we offer the following summary of our conclusions:

CONCLUSIONS

Let a_n be the first nonzero coefficient in Equation (4.2):

$$a_2 = a_3 = \cdots = a_{n-1} = 0.$$

Then our results fall into two cases, depending upon whether n is even or odd:

n even: Our results apply to systems in which all the coefficients b_i in Equation (4.1) are zero for $i < (n+2)/2$:

$$b_2 = b_3 = \cdots = b_{n/2} = 0.$$

If $a_n \neq 0$, then in some neighborhood of the origin, the flow given by the system (4) is topologically equivalent to the flow given by the simplified system

$$\begin{aligned} x' &= y, \\ y' &= a_n x^n. \end{aligned} \tag{6}$$

n odd: Our results apply to systems in which all the coefficients b_i in Equation (4.1) are zero for $i < (n+1)/2$:

$$b_2 = b_3 = \cdots = b_{(n-1)/2} = 0.$$

If $a_n \neq 0$ and if

$$4a_n + \frac{n+1}{2} b_{(n+1)/2}^2 > 0, \tag{7}$$

then in some neighborhood of the origin, the flow given by the system (4) is topologically equivalent to the flow given by the simplified system

$$\begin{aligned} x' &= y + b_{(n+1)/2}\, x^{(n+1)/2}, \\ y' &= a_n x^n. \end{aligned} \tag{8}$$

For example, consider the system

$$\begin{aligned} x' &= y + x^2 + x^3, \\ y' &= a_3 x^3 + x^4. \end{aligned} \tag{9}$$



FIG. 1a



FIG. 1b

FIG. 1. Numerical integration of systems (9) and (10) for $a_3 = 1$. (a) refers to (9), (b) to (10). Note similarity of flows, in agreement with our analytical results.

In the case that $a_3 > -\frac{1}{2}$ and $a_3 \neq 0$, the behavior of this system in a neighborhood of the origin is topologically equivalent to that of the truncated system:

$$x' = y + x^2,$$
$$y' = a_3 x^3. \tag{10}$$

In order to illustrate these conclusions, we compare numerical integrations of systems (9) and (10) in Figures 1, 2, and 3, which respectively correspond to the three values $a_3 = 1$, $-\frac{1}{5}$, and -1. For $a_3 = 1$ and $-\frac{1}{5}$, the systems (9) and (10) behave similarly near the origin. For $a_3 = -1$, however, the relation



FIG. 2a



FIG. 2b

FIG. 2. Numerical integration of systems (9) and (10) for $a_3 = -\frac{1}{5}$ (a) refers to (9), (b) to (10). Although globally these flows are quite different [e.g. (9) possesses a saddle at $x = 0.2$, $y = -0.048$], note their similarity in the neighborhood of the origin, in agreement with our analytical results. The uncompleted trajectories near the origin approach it asymptotically.

(7) fails to hold, and so our results do not guarantee local topological equivalence of (9) and (10), as is borne out by Figure 3.

TAKENS'S METHOD

Takens's method involves a study of the invariant manifolds (or separatrices) which pass through the equilibrium point at the origin. The limiting directions of approach of trajectories to the origin are obtained by transforming to polar coordinates

$$x = r\cos s, \qquad y = r\sin s \tag{11}$$



FIG. 3a



FIG. 3b

FIG. 3. Numerical integration of the systems (9) and (10) for $a_3 = -1$. (a) refers to (9), (b) to (10). In this case the system (9) is not determined by the truncated system (10).

and finding the equilibria of the s' equation in the limit as r approaches 0. If the flow in the neighborhood of these equilibria (in r, s coordinates) is uninfluenced by higher order terms (i.e. is determined), then so will be the topology of the invariant manifolds in the original equation. The process of transforming to polar coordinates and Taylor expanding about an equilibrium point of the s' equation is called "blowing up" the original singular point.

The blowing up process thus replaces the original flow (in x, y coordinates) with another (in r, s coordinates), about which the same question of determinacy is to be asked. The answer is given by a computation which we call Takens's test, which is equivalent to blowing up the newly derived singular point and testing for hyperbolicity.

If Takens's test does not result in a proof of determinacy, we may repeat the entire process on the newly derived equations (i.e. the equations in r, s

coordinates). Blowing up this flow will produce yet another derived flow which we again test for determinacy, etc. The process continues in this fashion until one of the following situations occurs:

(a) the new flow passes Takens's test, i.e. is determined, or

(b) a flow results which has no invariant manifolds through the origin (e.g. a generalized center or focus), in which case the method fails, or

(c) the algebra becomes too complicated to proceed. This practical limitation occurs even when using computer algebra. It happens, e.g. if the s' equation is so complicated that we cannot determine its equilibria analytically.

In what follows we describe the foregoing process in more detail, and then show how we implemented it on MACSYMA.

BLOWUP TRANSFORMATIONS

We take (4) in the form (1), and transform to polar coordinates (11), yielding

$$\begin{aligned} r' &= P(r,s) = \frac{f(r\cos s, r\sin s)}{r}, \\ s' &= Q(r,s) = \frac{g(r\cos s, r\sin s)}{r^2}, \end{aligned} \tag{12}$$

where

$$\begin{aligned} f(x,y) &= xu(x,y) + yv(x,y), \\ g(x,y) &= xv(x,y) - yu(x,y). \end{aligned} \tag{13}$$

Next we divide out any common factors of powers of r (say r^m) shared by P and Q. This step is equivalent to reparametrizing time with $dT = r^m dt$, and does not affect the shape of the integral curves of the system $dr/ds = P/Q$. We call the resulting functions PP and QQ:

$$\mathrm{PP}(r,s) = \frac{P(r,s)}{r^m}, \qquad \mathrm{QQ}(r,s) = \frac{Q(r,s)}{r^m}. \tag{14}$$

Since (1) possesses an equilibrium at $x = y = 0$, the equation $r' = \mathrm{PP}(r,s)$ has an equilibrium at $r = 0$, i.e. $\mathrm{PP}(0,s) = 0$.

We look for the limiting directions of approach of trajectories to the origin by requiring s' to vanish as r approaches zero. That is, from (12.2) and (14), we set

$$QQ(0, s) = 0 \tag{15}$$

Any solution $s = s^*$ to (15) represents (together with $r = 0$) an equilibrium solution of (12). We "blow up" the system in the neighborhood of this equilibrium by expanding PP and QQ in Taylor series about ($r = 0$, $s = s^*$), obtaining the new system

$$\begin{aligned} x_2' &= u_2(x_2, y_2), \\ y_2' &= v_2(x_2, y_2), \end{aligned} \tag{16}$$

where $x_2 = r$, $y_2 = s - s^*$, and where u_2 and v_2 are Taylor expansions of PP and QQ about $r = 0$, $s = s^*$, i.e. about $x_2 = y_2 = 0$.

In order to decide whether (16) is determined by its lowest degree terms, we use Takens's test, to be described next. If each system (16) which corresponds to a distinct choice of s^* passes Takens's test, i.e. is determined by its lowest degree terms, then the original system (1) is also determined. This follows because (16) and (1) are related by the continuous transformation (11).

On the other hand, if one of the systems (16) (corresponding to a particular equilibrium value s^*) fails Takens's test, we may view (16) as a new system of the form (1), and we may reapply the preceding algorithm by again transforming to polar coordinates and blowing up the equilibrium point. This will give a new system

$$\begin{aligned} x_3' &= u_3(x_3, y_3), \\ y_3' &= v_3(x_3, y_3), \end{aligned} \tag{17}$$

and so on. If after a finite number of such blowups all resulting systems pass Takens's test, then the original system (1) is determined.

TAKENS'S TEST

In order to prove his result for the case $a_2 \neq 0$, Takens used a theorem (p. 66 of his paper) which we call Takens's test. It concerns systems of the form

$$x' = u^*(x, y) + \cdots,$$
$$y' = v^*(x, y) + \cdots, \tag{18}$$

where u^* and v^* are homogeneous polynomials of degree k, and where $\cdots$ stands for polynomial terms of higher degree. Takens's test gives sufficient conditions on u^* and v^* such that the local flow of (18) is topologically equivalent to the local flow of the truncated system

$$x' = u^*(x, y),$$
$$y' = v^*(x, y). \tag{19}$$

The conditions are stated in terms of the functions f^* and g^*, where

$$f^*(x, y) = xu^*(x, y) + yv^*(x, y),$$
$$g^*(x, y) = xv^*(x, y) - yu^*(x, y). \tag{20}$$

Here f^* and g^* are homogeneous polynomials of degree $k+1$.

To apply Takens's test we solve $g^*(x, y) = 0$ and let p represent a real nontrivial solution (x, y). If

(i) $f^*(p) \neq 0$ and
(ii) $\operatorname{grad} g^*(p) \neq 0$

for all nontrivial solutions p, then the local behavior of (18) is determined by (19). In such a case we say that Takens's test has been passed. If either of conditions (i) and (ii) does not hold, or if there are no real nontrivial solutions p to $g^* = 0$, then no determinacy conclusion can be drawn, and we say that Takens's test has been failed.

Although we do not prove this theorem here, we note that it may be derived by blowing (19) up, dividing out common factors of r, and showing that when conditions (i) and (ii) hold, the new flow has hyperbolic equilibria, i.e. equilibria having eigenvalues with nonzero real parts, and which therefore are structurally stable (see Guckenheimer and Holmes for a discussion of structural stability).

Note that when we apply Takens's test to systems such as (16) or (17), we must insist on the additional requirement that $g^*(x, y) = 0$ have a real nontrivial root other than $x = 0$. This is because in a system coming from a blowup, like (16) or (17), $x=0$ corresponds to $r=0$ and thus does not represent an invariant manifold through the origin. This condition excludes centers and foci, to which Takens's method does not apply; cf. Figure 3.

EXAMPLE

As an example of this process, we consider the system

$$\begin{aligned} x' &= y + b_2 x^2 + b_3 x^3 + \cdots, \\ y' &= a_3 x^3 + a_4 x^4 + \cdots. \end{aligned} \tag{21}$$

Note that the system (9), illustrated in Figures 1–3, is a special case of this system. In the Appendix we offer a sample run of our MACSYMA program applied to the system (21).

Transforming to polar coordinates as in (11)–(13), we find

$$\begin{aligned} r' &= r\cos s \sin s + b_2 r^2 \cos^3 s + b_3 r^3 \cos^4 s + a_3 r^3 \cos^3 s \sin s + a_4 r^4 \cos^4 s \sin s, \\ s' &= -\sin^2 s - b_2 r \cos^2 s \sin s - b_3 r^2 \cos^3 s \sin s + a_3 r^2 \cos^4 s + a_4 r^3 \cos^5 s \end{aligned} \tag{22}$$

Next we set $r=0$ and look for equilibria of (22.2). This gives $s^*=0$ and π. Expanding (22) about ($r=0$, $s=0$), we obtain

$$\begin{aligned} r' &= b_2 r^2 + rs + \cdots, \\ s' &= a_3 r^2 - b_2 rs - s^2 + \cdots. \end{aligned} \tag{23}$$

Next we submit (23) to Takens's test. We rewrite (23) replacing r by x_2 and s by y_2:

$$\begin{aligned} x_2' &= b_2 x_2^2 + x_2 y_2 + \cdots, \\ y_2' &= a_3 x_2^2 - b_2 x_2 y_2 - y_2^2 + \cdots. \end{aligned} \tag{24}$$

To show that (24) is determined by its quadratic terms, we compute

$$\begin{aligned} f^* &= b_2 x_2^3 + (a_3+1) x_2^2 y_2 - b_2 x_2 y_2^2 - y_2^3, \\ g^* &= a_3 x_2^3 - 2b_2 x_2^2 y_2 - 2x_2 y_2^2. \end{aligned} \tag{25}$$

The solutions p to $g^* = 0$ are of the form

$$\frac{x_2}{y_2} = \frac{b_2 \pm \left(b_2^2 + 2a_3\right)^{1/2}}{a_3}, \tag{26}$$

as well as $x_2 = 0$. These values for p satisfy conditions (i) and (ii), except when $a_3 = 0$ or $a_3 \leqslant -b_2^2/2$.

When $a_3 = 0$, the equation $g^* = 0$ has the solutions $x_2/y_2 = b_2$, $x_2 = 0$, and $y_2 = 0$. The first of these gives $f^* = 0$, failing Takens's test for determinacy.

When $a_3 = -b_2^2/2$, the equation $g^* = 0$ has the solutions $x_2/y_2 = 2/b_2$ and $x_2 = 0$. The first of these gives grad $g^* = 0$, again failing Takens's test.

When $a_3 < -b_2^2/2$, there are no real roots (26) to $g^* = 0$ other than $x_2 = 0$, and Takens's test again fails to prove determinacy.

Thus we see that if $a_3 \neq 0$ and $a_3 > -b_2^2/2$, then the system (21) will be determined by the truncated system (8) for $n = 3$.

COMPUTER ALGEBRA

Although a single blowup transformation sufficed for a study of the determinacy of (21), in general many such transformations are required for treating a system of the form (4) when $a_2 = a_3 = \cdots = a_{n-1} = 0$. Moreover, each transformation typically involves several equilibria $s = s^*$, each of which must be investigated separately. This task involves a great deal of algebra and is best accomplished using a computer algebra system such as MACSYMA (see Rand [4] for an introduction to MACSYMA).

We wrote a MACSYMA program to perform the computations involved in Takens's method. The program is listed in the Appendix together with a sample run corresponding to the system (21). The computation corresponding to the ith blowup may be outlined as follows:

1. Given $u_i(x_i, y_i)$ and $v_i(x_i, y_i)$, compute f_i and g_i from (13).
2. Perform Takens's test:
   a. Truncate f_i and g_i to homogeneous polynomials.
   b. Set $g_i = 0$ and solve for roots.
   c. Evaluate f_i, and the partial derivatives g_{ix} and g_{iy}, for each root.
   d. If f_i and either of g_{ix} or g_{iy} are not zero for each root, you are done. Otherwise continue with
3. Compute P_i and Q_i from (12).

TABLE 1

RESULTS OF COMPUTER ALGEBRA INVESTIGATION

First nonzero coefficients in (4)	Conditions(s) for determinacy	No. of blowups required
a_2, b_2	$a_2 \neq 0$	2
a_3, b_2	$a_3 \neq 0, 4a_3 + 2b_2^2 > 0$	1
a_4, b_3	$a_4 \neq 0$	3
a_5, b_3	$a \neq 0, 4a_5 + 3b_3^2 > 0$	2
a_6, b_4	$a_6 \neq 0$	4
a_7, b_4	$a_7 \neq 0, 4a_7 + 4b_4^2 > 0$	3
a_8, b_5	$a_8 \neq 0$	5
a_9, b_5	$a_9 \neq 0, 4a_9 + 5b_5^2 > 0$	4

4. Compute PP_i and QQ_i by dividing out common powers of r_i from P_i and Q_i [see (14)].
5. Set $r_i = 0$ in QQ_i and solve for $s_i = s_i^*$.
6. For each solution s_i^*, compute u_{i+1} and v_{i+1} by Taylor expanding PP_i and QQ_i about $x_{i+1} = r_i = 0$, $y_{i+1} = s_i - s_i^* = 0$.
7. Go to step 1.

Using the MACSYMA program given in the Appendix, we applied this algorithm to a variety of systems of the form (4). In some cases, the algebra became too involved to draw any conclusions. Such was the case, e.g. when $a_2 = a_3 = 0$, $a_4 \neq 0$, and $b_2 \neq 0$.

For the class of systems discussed previously, however, the method worked well. We ran cases displayed in Table 1, from which we drew the conclusions stated in the discussion of Equations (6)–(8). Thus we cannot prove those assertions for all n, but from the form of the results in Table 1, they seem very likely to be true.

APPLICATION TO THE VAN DER POL EQUATION

The present work was stimulated by an application to the behavior of van der Pol's equation at infinity [2]. If in van der Pol's equation

$$w'' + w - ew'(1 - w^2) = 0, \tag{27}$$

we set (see [3], p. 92)

$$u = w/w' \quad \text{and} \quad z = 1/w' \tag{28}$$

and reparametrize time with $dT = w'^2\, dt$. We obtain

$$u' = z^2(1+u^2-eu)+eu^3,$$
$$z' = uz^3 + ez(u^2-z^2). \tag{29}$$

The origin $u = z = 0$ is a singular point at infinity. We are interested in the nature of the local behavior in the neighborhood of this point. Note that the flow (29) has no linear part. However, by setting

$$v = z^2 \tag{30}$$

we obtain

$$u' = v - euv + eu^3 + u^2 v,$$
$$v' = -2ev^2 + 2uv^2 + 2eu^2 v. \tag{31}$$

Although (31) has the linear part which we have considered in this paper, the nonlinear terms are not in the normal form (4). By using the theory of normal forms [1] and MACSYMA [5], we were able to show that the near-identity transformation

$$u = x - \tfrac{3}{2}ex^2 + \frac{4+15e^2}{6}x^3 - \frac{70e+105e^3}{24}x^4$$
$$+\frac{88+1086e^2+945e^4}{120}x^5,$$
$$v = y - 2exy + \tfrac{2}{3}ex^3 + (1+4e^2)x^2y - \tfrac{7}{3}ex^4$$
$$-(5e+8e^3)x^2y + \frac{24e+185e^3}{30}x^5 + \frac{16+207e^2+192e^4}{12}x^4y \tag{32}$$

takes (31) into the form

$$x' = y + \tfrac{5}{3}ex^3 - \tfrac{5}{2}e^2x^4 + \frac{8e+225e^3}{60}x^5 + O(6),$$
$$y' = -2e^2x^5 + O(6). \tag{33}$$

The system (33) is of the form (4) with $b_3 = \frac{5}{3}e$ and $a_5 = -2e^2$. For these parameter values, the relation (7) holds,

$$4a_5 + 3b_3^2 = e^2/3 > 0, \tag{34}$$

and so we may conclude that in some neighborhood of the origin, the flow (33) is topologically equivalent to the flow of the simplified system

$$\begin{aligned} x' &= y + \tfrac{5}{3}ex^3, \\ y' &= -2e^2x^5. \end{aligned} \tag{35}$$

Note that without this result one could not safely truncate the system (33) at, say, $O(6)$ terms, since it might have been the case that such terms determined the topological nature of the local flow.

APPENDIX

We list below the MACSYMA program we used in this investigation, followed by a sample run corresponding to the system (21) discussed previously:

```
/* TAKENS' BLOW-UP CALCULATION */

/* MAIN PROGRAM */
TAKENS( ):=BLOCK(
/* VARIABLE I TAGS LOOP */
FOR I:1 THRU 8 DO
      (SETUP1( ),
      IF I=1 THEN INPUTRHS( ),
      IF I>1 THEN BLOWUP( ),
      SETUP2( ),
      DEFFG( ),
      PRINT(" TAKENS' TEST "),
      PRINT(" TRUNCATE F AND G TO HOMOGENEOUS POLYNOMIALS "),
      PRINT(TRUNCATE( )),
      GETROOTS( ),
      PRINT(TEST( )),
      IF FLAG=GREEN THEN RETURN("DONE"),
      DEFPQ( ),
      DEFPPQQ( ),
      SROOTS( )) )$

/* SUBROUTINES TO CREATE VARIABLE NAMES AT ITH LOOP */
SETUP1( ):=(
U:CONCAT('U,I),
V:CONCAT('V,I),
X:CONCAT('X,I),
Y:CONCAT('Y,I))$

SETUP2( ):=(
F:CONCAT('F,I),
G:CONCAT('G,I),
```

```
P:CONCAT('P,I),
Q:CONCAT('Q,I),
R:CONCAT('R,I),
S:CONCAT('S,I),
PP:CONCAT('PP,I),
QQ:CONCAT('QQ,I))$

/* SUBROUTINE TO INPUT THE RHS'S FROM KEYBOARD */
INPUTRHS( ):=(
PRINT(" ENTER THE RHS'S TO BE STUDIED "),
PRINT(" USE VARIABLES X,Y, THEY WILL BE CONVERTED TO X1,Y1 "),
U::READ(U,"="),
PRINT(EV(U)),
V::READ(V,"="),
PRINT(EV(V)))$

/* SUBROUTINE TO TRUNCATE F AND G TO TERMS OF LOWEST DEGREE */
TRUNCATE( ):=BLOCK(
FOR J FROM 2 THRU 8 DO
      (TEMP1:RATEXPAND([EV(F),EV(G)]),
      TEMP2:TAYLOR(TEMP1,[EV(X),EV(Y)],0,J),
      IF TEMP2 # TAYLOR([0,0],DUMMY,0,1) THEN RETURN(TEMP2)))$

/* SUBROUTINE TO SOLVE GTRUNC = 0 */
GETROOTS( ):=(
PRINT("SOLVING GTRUNC = 0"),
FTRUNC:PART(TEMP2,1),
GTRUNC:PART(TEMP2,2),
GTRUNCX:DIFF(GTRUNC,EV(X)),
GTRUNCY:DIFF(GTRUNC,EV(Y)),
XROOTS:SOLVE(GTRUNC,EV(X)),
YROOTS:SOLVE(GTRUNC,EV(Y)),
ROOTNUM:0,
FOR K:1 THRU LENGTH(XROOTS) DO
     (ROOTNUM:ROOTNUM+1,
     ROOT[ROOTNUM]:PART(XROOTS,K)),
FOR K:1 THRU LENGTH(YROOTS) DO
     (ROOTNUM:ROOTNUM+1,
     ROOT[ROOTNUM]:PART(YROOTS,K)),
PRINT("TOTAL NO. OF ROOTS =",ROOTNUM))$

/* PERFORM TAKENS' TEST FOR EACH ROOT */
TEST( ):=(BLOCK(
FLAG:GREEN,
FOR K:1 THRU ROOTNUM DO (PRINT(ROOT[K]),
     FTEST:EV(FTRUNC,ROOT[K]),
     GXTEST:EV(GTRUNCX,ROOT[K]),
     GYTEST:EV(GTRUNCY,ROOT[K]),
/*   PRINT("FTRUNC =",FTEST),
     PRINT("GXTRUNC =",GXTEST),
     PRINT("GYTRUNC =",GYTEST), */
     IF FTEST=0 THEN (PRINT("FTRUNC IS ZERO!"), FLAG:RED) ELSE
     IF GXTEST=0 AND GYTEST=0 THEN (PRINT("DG TRUNC IS
ZERO!"),FLAG:RED)),
     IF FLAG=GREEN THEN "PASSED TEST" ELSE "FAILED TEST"))$

/* SUBROUTINE TO DEFINE F AND G */
DEFFG( ):=(
F::EXPAND(EV(X*U+Y*V)),
PRINT(F,"=",EV(F)),
G::EXPAND(EV(X*V-Y*U)),
PRINT(G,"=",EV(G)))$

/* SUBROUTINE TO DEFINE P AND Q */
DEFPQ( ):=(
```

```
TRANS:[EV(X)=R*COS(S),EV(Y)=R*SIN(S)],
P::EV(F)/R,P::EXPAND(EV(EV(P),TRANS)),
PRINT(P,"=",EV(P)),
Q::EV(G)/R^2,Q::EXPAND(EV(EV(Q),TRANS)),
PRINT(Q,"=",EV(Q)))$

/* SUBROUTINE TO DEFINE PP AND QQ */
DEFPPQQ( ):=(
EXPONENT:MIN(LOPOW(EV(P),R),LOPOW(EV(Q),R)),
PP::EXPAND(EV(P)/R^EXPONENT),
QQ::EXPAND(EV(Q)/R^EXPONENT),
PRINT("DIVIDE OUT",R^EXPONENT),
/*     PRINT(PP,"=",EV(PP)),
       PRINT(QQ,"=",EV(QQ)),      */
PRINT("NOW SET",R,"=0"),
PTEMP:EV(EV(PP),R::0),
QTEMP:EV(EV(QQ),R::0),
PRINT(PP,"=",PTEMP),
PRINT("NOTE: PREVIOUS SHOULD BE ZERO!"),
PRINT(QQ,"=",QTEMP))$

/* SUBROUTINE TO FIND ROOTS S OF QQ=0 WHEN R=0     */
/*     USER SELECTS ROOT SSTAR TO BE USED          */
SROOTS( ):=(
STEMP:SOLVE(QTEMP,EV(S)),
FOR K:1 THRU LENGTH(STEMP) DO
     PRINT("ROOT NO.",K,",",PART(STEMP,K)),
     PRINT("THERE ARE",LENGTH(STEMP),"ROOTS"),
     ROOTNO:READ("PICK A ROOT NO., OR 0 TO ENTER ONE"),
     IF ROOTNO=0 THEN SSTAR:READ("ENTER ROOT")ELSE
     SSTAR:RHS(PART(STEMP,ROOTNO)),
     PRINT(S,"STAR =",SSTAR))$

/* SUBROUTINE TO TAYLOR EXPAND PP AND QQ ABOUT R=0, S=STAR */
/*        RETURNS NEW U AND V FOR NEXT ITERATION          */

BLOWUP( ):=(
R::EV(X),
S::SSTAR+EV(Y),
POW:READ("KEEP TERMS OF WHAT POWER?"),
PRINT(U,"="),
U::TAYLOR(EV(EV(PP)),[EV(X),EV(Y)],0,POW),
PRINT(EV(U)),
PRINT(V,"="),
V::TAYLOR(EV(EV(QQ)),[EV(X),EV(Y)],0,POW),
PRINT(EV(V)) )$
```

Sample Run

This run corresponds to the example (21) given in the text. Lines ending in a semicolon have been entered from the keyboard by the user.

```
(C21) TAKENS( );
 ENTER THE RHS'S TO BE STUDIED
 USE VARIABLES X, Y, THEY WILL BE CONVERTED TO X1, Y1
U1 =
Y+B2*X^2+B3*X^3;
           3        2
Y1 + B3 X1   + B2 X1
```

```
V1 =
A3*X^3+A4*X^4;
     4        3
A4 X1  + A3 X1
          4           3                   4        3
F1 = A4 X1  Y1 + A3 X1  Y1 + X1 Y1 + B3 X1  + B2 X1
         2        3           2           5        4
G1 = - Y1  - B3 X1  Y1 - B2 X1  Y1 + A4 X1  + A3 X1
 TAKENS' TEST
 TRUNCATE F AND G TO HOMOGENEOUS POLYNOMIALS
                    2
[Y1 X1 + . . ., - Y1 + . . .]
SOLVING GTRUNC = 0
TOTAL NO. OF ROOTS = 1
Y1 = 0
FTRUNC IS ZERO!
FAILED TEST
          4    4                    3    3
P1 = A4 R1  COS (S1) SIN(S1) + A3 R1  COS (S1) SIN(S1)

                              3    4            2    3
  + R1 COS(S1) SIN(S1) + B3 R1  COS (S1) + B2 R1  COS (S1)
          2            2    3
Q1 = - SIN (S1) - B3 R1  COS (S1) SIN(S1)

            2                    3    5
 - B2 R1 COS (S1) SIN(S1) + A4 R1  COS (S1)

        2    4
 + A3 R1  COS (S1)
DIVIDE OUT 1
NOW SET R1 = 0
PP1 = 0
NOTE: PREVIOUS SHOULD BE ZERO!
           2
QQ1 = - SIN (S1)

SOLVE is using arc-trig functions to get a solution.
Some solutions may be lost.
ROOT NO. 1 , S1 = 0
THERE ARE 1 ROOTS
PICK A ROOT NO., OR 0 TO ENTER ONE
1;
S1 STAR = 0
KEEP TERMS OF WHAT POWER?
3;
```

$$U2 = B2\,X2^{2} + Y2\,X2 + B3\,X2^{3} + \ldots$$

$$V2 = A3\,X2^{2} - B2\,Y2\,X2 - Y2^{2} + (A4\,X2^{3} - B3\,Y2\,X2^{2}) + \ldots$$

$$F2 = - Y2^{3} - B3\,X2^{2}\,Y2^{2} - B2\,X2\,Y2^{2} + A4\,X2^{3}\,Y2 + A3\,X2^{2}\,Y2 + X2^{2}\,Y2 + B3\,X2^{4} + B2\,X2^{3}$$

$$G2 = - 2\,X2\,Y2^{2} - 2\,B3\,X2^{3}\,Y2 - 2\,B2\,X2^{2}\,Y2 + A4\,X2^{4} + A3\,X2^{3}$$

TAKENS' TEST

TRUNCATE F AND G TO HOMOGENEOUS POLYNOMIALS

$$[B2\,X2^{3} + (A3 + 1)\,Y2\,X2^{2} - B2\,Y2^{2}\,X2 - Y2^{3} + \ldots,\quad A3\,X2^{3} - 2\,B2\,Y2\,X2^{2} - 2\,Y2^{2}\,X2 + \ldots]$$

SOLVING GTRUNC = 0

TOTAL NO. OF ROOTS = 5

$$X2 = \frac{B2\,Y2 - \mathrm{SQRT}(B2^{2} + 2\,A3)\,Y2}{A3}$$

$$X2 = \frac{\mathrm{SQRT}(B2^{2} + 2\,A3)\,Y2 + B2\,Y2}{A3}$$

$$X2 = 0$$

$$Y2 = - \frac{\mathrm{SQRT}(B2^{2} + 2\,A3)\,X2 + B2\,X2}{2}$$

$$Y2 = - \frac{B2\,X2 - \mathrm{SQRT}(B2^{2} + 2\,A3)\,X2}{2}$$

PASSED TEST

(D21) DONE

REFERENCES

1 J. Guckenheimer and P. Holmes, *Nonlinear Oscillations, Dynamical Systems, and Bifurcations of Vector Fields*, Springer, 1983.
2 W. L. Keith and R. H. Rand, Dynamics of a system exhibiting the global bifurcation of a limit cycle at infinity, *Internat. J. Non-Linear Mech.*, to appear.
3 N. Minorsky, *Nonlinear Oscillations*, Van Nostrand, 1962.
4 R. H. Rand, *Computer Algebra in Applied Mathematics: An Introduction to* MACSYMA, Pitman, 1984.
5 R. H. Rand, and W. L. Keith, Normal form and center manifold calculations on MACSYMA, in *Applications of Computer Algebra*, (R. Pavelle, Ed.), to appear.
6 F. Takens, Singularities of vector fields, *Publ. Math. Inst. Hautes Etudes Sci.* 43:47–100 (1974).